

# CMPT 360: Lab Assignment #5

## Minesweeper

Brady Coles

last updated: January 3, 2018

### Contents

<b>1</b>	<b>Course Goals</b>	<b>2</b>
<b>2</b>	<b>Problem Description</b>	<b>2</b>
<b>3</b>	<b>Program Documentation</b>	<b>2</b>
3.1	Description . . . . .	2
3.2	Documentation . . . . .	2
<b>4</b>	<b>Program Listing</b>	<b>3</b>
4.1	Main Program . . . . .	3
4.2	Form Definition . . . . .	7
4.3	Button Component . . . . .	8
<b>5</b>	<b>Sample Operation</b>	<b>9</b>

# 1 Course Goals

This assignment fulfills the following goals:

- a group II language (Delphi)
- implemented on the Windows platform

## 2 Problem Description

The given problem is to create a playable single player game. The described program is a recreation of Minesweeper for the windows platform, including a fully functional graphical user interface.

## 3 Program Documentation

### 3.1 Description

The minesweeper game is made up of a grid of tiles, where some portion of the tiles are mines. The player's goal is to reveal all tiles except the mines, and when a tile is revealed, it displays the number of mines in the eight tiles it touches. Using these numbers, the player can determine where the mines are and are not.

In this program, the user can start a new game anytime, resetting the game board and randomizing the location of the mines. The board is 9x9, with ten mines, the standard 'easy' board in Microsoft minesweeper. Any tiles not yet revealed can be flagged, and flagged tiles can not be revealed without first removing the flag. When a tile touching zero mines is revealed, all tiles it touches is also revealed.

When a mine is revealed, the game ends in a loss; when all tiles but the mines are revealed, the game ends in a win. A message displaying the nature of the end of the game is shown when the game ends. During a game, this message displays the number of mines on the board.

A cell is revealed by left clicking on the tile. Left clicking a revealed or flagged tile does nothing. The flag on a tile is toggled by right clicking on the tile. Right clicking on a revealed tile does nothing. Flagging mines has no effect on winning or losing, it is simply a tool and safeguard for the user to use to improve play.

### 3.2 Documentation

Certain program units can be utilized in other programs.

The unit `GridButton` contains the class `TGridButton` which inherits from the class `Vcl.StdCtrls.TButton`. Two properties are added, `X` and `Y`, which can be used to determine the location of a button within a grid.

The main program is not extensible, nor is its unit usable in other programs, unless it is used as a separate window.

The main program is defined in two parts, the program definition and the GUI (form) definition. The form definition is auto generated by the Delphi editor, RAD studio, and defines the objects in the window at program startup. The game grid is not generated until runtime.

The record type `CellContent` holds data for an individual cell, namely three boolean fields, `isMine` is true if the cell is a mine, `isVisible` is true if the cell has been revealed, and `isFlagged` is true if the cell has been flagged by the user.

The array types `GridOfCells` and `GridOfButtons` are for holding the data and button objects of a game board, respectively.

The constants are:

- `TOP_BUFFER` Pixels above the game board.
- `SIDE_BUFFER` Pixels to each side of the game board.
- `BOTTOM_BUFFER` Pixels below the game board.
- `CELL_SIZE` The pixel length of a side of a cell.

## 4 Program Listing

### 4.1 Main Program

```
1  // Author: Brady Coles
2  // Lab Assignment # 5
3  // Minesweeper clone
4  unit Minesweeper;
5
6  interface
7
8  uses
9      Winapi.Windows, Winapi.Messages, System.SysUtils, System.Variants, System.Classes, Vcl.Graphics,
10     Vcl.Controls, Vcl.Forms, Vcl.Dialogs, Vcl.StdCtrls, GridButton, Vcl.ExtCtrls, Math;
11
12 type
13     CellContent = record
14         isMine, isVisible, isFlagged : Boolean;
15     end;
16     Whole = 1..MaxInt;
17     GridOfCells = array[1..9, 1..9] of CellContent;
18     GridOfButtons = array[1..9, 1..9] of TGridButton;
19     TForm1 = class(TForm)
20         GridButton1 : TGridButton;
21         StatusLabel: TLabel;
22         Label1: TLabel;
23         procedure NewGameClick(Sender: TObject);
24         procedure GridButtonClick(Sender: TObject; Button: TMouseButton; Shift: TShiftState;
25             PixelX, PixelY: Integer);
26     private
27         { Private declarations }
28     public
29         { Public declarations }
30     end;
31
32 const
33     TOP_BUFFER = 60;
34     SIDE_BUFFER = 25;
35     BOTTOM_BUFFER = 60;
36     CELL_SIZE = 25;
37 var
38     Form1: TForm1;
39     Cells: GridOfCells;
40     Buttons: GridOfButtons;
41     BoardWidth, BoardHeight, NumberOfMines : Whole;
42
43 implementation
44
45 {LR *.dfm} // Include form definition
46
47 // Returns true if coordinates refer to a valid cell, false otherwise.
48 function IsValidCell(x, y: Integer): Boolean;
49 begin
50     Result := (x >= 1) and (x <= BoardWidth) and (y >= 1) and (y <= BoardHeight);
```

```

51 end;
52
53 // Returns number of mines around a cell. Amount included cell (if cell is a mine).
54 function CellMineCount(x, y: Whole): Integer;
55 var
56     i, j : Whole;
57 begin
58     Result := 0;
59     if IsValidCell(x, y) then
60         for i := (x - 1) to (x + 1) do
61             for j := (y - 1) to (y + 1) do
62                 if IsValidCell(i, j)
63                     then if Cells[i][j].isMine then Result := Result + 1;
64 end;
65
66 // Reveals the cell, giving the surrounding mine count if not a mine, or
67 // the '*' symbol if the cell is a mine. Also disables the button. If the cell
68 // has zero mines around it, reveals all surrounding cells as well (recursively).
69 procedure Reveal(x, y: Whole);
70 var
71     i, j, count : Integer;
72 begin
73     if IsValidCell(x, y) and not Cells[x][y].isVisible then
74         begin
75             if Cells[x][y].isMine then
76                 begin
77                     Buttons[x][y].Caption := '*';
78                     Buttons[x][y].Enabled := False;
79                     Cells[x][y].isVisible := True;
80                 end
81             else
82                 begin
83                     count := CellMineCount(x, y);
84                     Buttons[x][y].Caption := count.ToString;
85                     Buttons[x][y].Enabled := False;
86                     // Before recursive call to avoid infinite recursion.
87                     Cells[x][y].isVisible := True;
88                     if count = 0 then
89                         // Reveal surrounding cells.
90                         for i := (x - 1) to (x + 1) do
91                             for j := (y - 1) to (y + 1) do
92                                 if IsValidCell(i, j)
93                                     then Reveal(i, j);
94                     end;
95                 end;
96 end;
97
98 // Toggles the flag on a game cell. If the cell has not been revealed, flags
99 // the cell if not flagged, removes flag if flagged. Flagged cells cannot be
100 // revealed.
101 procedure Flag(x, y : Integer);
102 begin
103     if IsValidCell(x,y) and (not Cells[x][y].isVisible) then
104         begin

```

```

105     if Cells[x][y].isFlagged then
106     begin
107         Cells[x][y].isFlagged := False;
108         Buttons[x][y].Caption := '';
109     end
110     else
111     begin
112         Cells[x][y].isFlagged := True;
113         Buttons[x][y].Caption := 'F';
114     end;
115 end;
116 end;
117
118
119 // Initialized data for game board. Prepared data for window initialization.
120 // Randomly chooses cells to be mines.
121 procedure InitializeBoard(width, height, mines : Whole);
122 var
123     rand, x, y, i, j : Whole;
124 begin
125     BoardWidth := width;
126     BoardHeight := height;
127     NumberOfMines := mines;
128     // Initialize data, needed for resets between games.
129     for i := 1 to BoardWidth do
130         for j := 1 to BoardHeight do
131             begin
132                 Cells[i][j].isMine := False;
133                 Cells[i][j].isVisible := False;
134                 Cells[i][j].isFlagged := False;
135             end;
136         // Select mines
137         for i := 1 to NumberOfMines do
138             begin
139                 repeat
140                     rand := RandomRange(1, BoardWidth * BoardHeight);
141                     x := rand mod BoardWidth;
142                     y := rand div BoardHeight;
143                     until not Cells[x][y].isMine;
144                     Cells[x][y].isMine := True;
145                 end;
146             end;
147         end;
148     // Set up form for a game. Makes window proper size for game board, creates
149     // or resets game cells.
150     procedure InitializeWindow;
151     var
152         B : TGridButton;
153         i, j : Whole;
154     begin
155         // Set up window
156         Form1.Width := BoardWidth * CELL_SIZE + 2 * SIDE_BUFFER
157             + Form1.Margins.Left + Form1.Margins.Right + 10;
158         Form1.Height := BoardHeight * CELL_SIZE + TOP_BUFFER + BOTTOM_BUFFER

```

```

159     + Form1.Margins.Top + Form1.Margins.Bottom;
160 Form1.StatusLabel.Caption := format('%u Mines', [NumberOfMines]);
161 // Set up cells
162 for i := 1 to BoardWidth do
163     for j := 1 to BoardHeight do
164         begin
165             if (Buttons[i][j] = nil) then Buttons[i][j] := TGridButton.Create(Form1);
166             Buttons[i][j].X := i;
167             Buttons[i][j].Y := j;
168             Buttons[i][j].Height := CELL_SIZE;
169             Buttons[i][j].Width := CELL_SIZE;
170             Buttons[i][j].Left := SIDE_BUFFER + (i - 1) * CELL_SIZE;
171             Buttons[i][j].Top := TOP_BUFFER + (j - 1) * CELL_SIZE;
172             Buttons[i][j].Caption := '';
173             Buttons[i][j].Parent := Form1;
174             Buttons[i][j].OnMouseDown := Form1.GridButtonClick;
175             Buttons[i][j].Enabled := True;
176         end;
177     end;
178
179 // Checks if the game has been won. Returns true if only unclicked cells
180 // are mines.
181 function CheckWinState : Boolean;
182 var
183     i, j : Integer;
184 begin
185     for i := 1 to BoardWidth do
186         for j := 1 to BoardHeight do
187             if (not Cells[i][j].isVisible) and (not Cells[i][j].isMine) then
188                 begin
189                     Result := False;
190                     exit;
191                 end;
192             Result := True;
193         end;
194     end;
195 // Ends the game by revealing all cells and changing label to win or loss message
196 procedure EndGame(isWin : Boolean);
197 var
198     i, j : Whole;
199 begin
200     for i := 1 to BoardWidth do
201         for j := 1 to BoardHeight do
202             Reveal(i, j);
203         if isWin then Form1.StatusLabel.Caption := 'You Win!'
204         else Form1.StatusLabel.Caption := 'You Lose!'
205     end;
206
207 // Handle starting a new game. Creates a 9x9 game with 10 mines.
208 procedure TForm1.NewGameClick(Sender: TObject);
209 begin
210     InitializeBoard(9, 9, 10);
211     InitializeWindow;
212 end;

```

```

213
214 // Handle a click on a game cell.
215 // Checks mouse button, right is flag, left is reveal.
216 // Handles ending the game on a win or loss.
217 procedure TForm1.GridButtonClick(Sender: TObject; Button: TMouseButton; Shift: TShiftState;
218     PixelX, PixelY: Integer);
219 var
220     x, y : Whole;
221     count : Integer;
222 begin
223     // Ensure that the event was on a game cell
224     if Sender is TGridButton then
225     begin
226         x := TGridButton(Sender).X;
227         y := TGridButton(Sender).Y;
228         if not IsValidCell(x, y) then exit;
229         case Button of
230             // Reveal. Note that already revealed cells will be disabled, so all
231             // events should be on unrevealed cells. If cell is revealed, Reveal
232             // procedure will ignore it.
233             mbLeft:
234             begin
235                 if Cells[x][y].isFlagged then exit;
236                 if Cells[x][y].isMine then EndGame(False) // Game over, player clicked mine
237                 else
238                     // Reveal cell, if last cell, win game.
239                     begin
240                         Reveal(x, y);
241                         if CheckWinState then EndGame(True);
242                     end;
243                 end;
244                 // Flag or unflag cell, depending on current state.
245                 mbRight: Flag(x, y)
246             end;
247         end;
248     end;
249 end.
250

```

## 4.2 Form Definition

```

1  object Form1: TForm1
2      Left = 0
3      Top = 0
4      Caption = 'Minesweeper'
5      ClientHeight = 58
6      ClientWidth = 275
7      Color = clBtnFace
8      Font.Charset = DEFAULT_CHARSET
9      Font.Color = clWindowText
10     Font.Height = -11
11     Font.Name = 'Tahoma'
12     Font.Style = []
13     OldCreateOrder = False

```

```

14 PixelsPerInch = 96
15 TextHeight = 13
16 object StatusLabel: TLabel
17     Left = 147
18     Top = 29
19     Width = 70
20     Height = 13
21 end
22 object Label1: TLabel
23     Left = 24
24     Top = 5
25     Width = 152
26     Height = 13
27     Caption = 'Brady Coles - CMPT 360 Lab #5'
28 end
29 object GridButton1: TGridButton
30     Left = 24
31     Top = 24
32     Width = 91
33     Height = 25
34     Caption = 'Start New Game'
35     TabOrder = 0
36     OnClick = NewGameClick
37     X = 0
38     Y = 0
39 end
40 end

```

### 4.3 Button Component

```

1  unit GridButton;
2
3  interface
4
5  uses
6      System.SysUtils, System.Classes, Vcl.Controls, Vcl.StdCtrls;
7
8  type
9      TGridButton = class(TButton)
10     private
11         FX, FY : Integer;
12     protected
13         { Protected declarations }
14     public
15         { Public declarations }
16     published
17         property X: Integer read FX write FX;
18         property Y: Integer read FY write FY;
19     end;
20
21 procedure Register;
22
23 implementation
24

```



```


25 procedure Register;
26 begin
27     RegisterComponents('Samples', [TGridButton]);
28 end;
29
30 end.

```


## 5 Sample Operation

Here are some screenshots of the program at different states.


**After Startup** The window has no game board, just a label and a start game button.




**New Game** The window resizes to fit the game board, and displays the number of mines.



**Game in Progress** An in progress game with hidden, revealed, and flagged cells.



**Game Over: Loss** A lost game, revealing all tiles, and displaying the message 'You Lose!'



**Game Over: Win** A won game, revealing all tiles, and displaying the message 'You Win!'

